

GB010: 3次元の熱伝導


本 whitepaper は Gunnar Backstrom 氏の承諾のもと、書籍“*Simple Fields of Physics by Finite Element Analysis*” に記されている多数の FlexPDE 適用事例の中からその一部を紹介するものです。



FlexPDE で 3 次元の場を扱う場合には、 x - y 平面に定義された base plane を z 軸方向に押出す (extrusion) という考え方が基本になります。

1 Case1 - 平衡状態

ここでは下図のような鉄の板に銅製の円柱を溶接した構造体 (単位は m) を考え、その円柱部上面に $1e6$ の熱流束を与えたときの平衡状態における熱伝導の様子を FlexPDE を用いて解析します。ただし境界条件としては次ページの平面図、立面図に示したようなものを想定します。

なお、支配方程式に関する背景情報については GB004 を参照ください。



平面図	立面図
	
<p>正面は熱的に絶縁されており、他は一定温度 300°K に保つものとしします。この条件は z 軸方向に extrude された側壁に対しても適用される点に注意してください。</p>	<p>底面は熱的に絶縁状態、一方円柱部上面からは $1e6$ の熱流束が与えられるものとしします。</p>

1.1 Problem descriptor [3dfields2a.pde]

まずタイトルを設定します。

```
TITLE
'Steady Heat Conduction' { 3dfields2a.pde }
```

次に演算精度に関するセレクトをセットします。

```
SELECT
Errlim = 1e-3
```

次に座標系が 3 次元直交座標系であることを明示します (デフォルトは Cartesian2)。

```
COORDINATES
Cartesian3
```

従属変数を定義します。

```
VARIABLES
temp
```

関連するパラメータや数式を定義します。 r_0 は円柱の半径、中心の位置は $(L/4, L/4)$ 、 z_0 は鉄板の厚さを意味しています。

DEFINITIONS

```
L = 0.1  r0 = 0.01  z0 = 0.02
k  heat = 0          { Thermal conductivity and power density }
in_fluxd = 1e6      { Input heat flux density at top }
fluxd_x = -k*dx(temp)  fluxd_y = -k*dy(temp)  fluxd_z = -k*dz(temp)
fluxd = vector(fluxd_x, fluxd_y, fluxd_z)
fluxdm = magnitude(fluxd)
```

3次元の熱伝導方程式を定義します。この偏微分方程式の誘導については GB004 を参照ください。

EQUATIONS

```
div(fluxd) = heat
```

z 軸方向への extrusion を定義します。下から順に境界面 (Surface) とそれらにはさまれた層 (Layer) を規定して行きます。

```
EXTRUSION { Parallel surfaces }
Surface 'bottom' z = 0
Layer 'iron'
Surface 'middle' z = z0 { Interface }
Layer 'copper'
Surface 'top' z = L
```

BOUNDARIES セクションにおいて base plane 上でのリージョン構成と境界条件を定義します (上記平面図参照)。これら base plane 上での形状定義や境界条件は基本的に z 軸方向に extrude されて行きます。ただし上面と下面については Surface 文にて境界条件を明示します。なおリージョン 'domain' の定義中、レイヤ 'copper' の部分については void という修飾子を用いている点に注意してください。リージョン 'cylinder' 内では鉄、銅双方の熱伝導率が定義されます。

BOUNDARIES

```

Surface 'bottom' Natural(temp) = 0      { Insulated }
Surface 'top' Natural(temp) = -in_fluxd { Input flux density }

Region 'domain'                        { Full solution domain }
  Layer 'iron' k = 82                   { Fe }
  Layer 'copper' void
  Start 'outer' (0, 0)
    Natural(temp) = 0 Line to (L, 0)
    Value(temp) = 300 Line to (L, L) to (0, L) to Close

Region 'cylinder'
  Layer 'iron' k = 82                    { Fe }
  Layer 'copper' k = 400                 { Redefine void to Cu }
  Start (L/4+r0, L/4) Arc(Center = L/4, L/4)
    Angle = 360 to Close

```

最後に出力すべき情報を規定します。

PLOTS

```

Grid(x, y, z)
Contour(temp) on x = y
Contour(temp) painted on x = y
Contour(fluxdm) painted on x = y
Vector(fluxd) norm on x = y
Contour(temp) on y = 0
Contour(temp) on y = L/4
Contour(temp) on x = L/4
Contour(temp) on z = z0 on 'iron'
Contour(temp) on z = 0


```

END

1.2 実行結果


(1) Grid(x, y, z)

FlexPDE によって生成された 3 次元のメッシュ構成を示しています。




(2) Contour(temp) on $x = y$

対角面 $x = y$ 上における等温線図を示したものです。銅製の円柱部は頂部が約 600°K で、下に向かってほぼリニアに温度は下がって行きます。また鉄製の基盤部両端は 300°K に保たれていることがわかります。




(3) Contour(temp) painted on $x = y$

プロット (2) と同一内容ですが、塗りつぶしを施した等温線図です。


(4) Contour(fluxdm) painted on $x = y$

熱流束密度ベクトル f の絶対値についての等高線図です。円柱頂部から $1e6$ の熱流束が流れ込んでいることを反映して、円柱内での $|f|$ はほぼ一様に $1e6$ という値になっています。なお、最大値は鉄板との接合部で発生しています。




(5) Vector(fluxd) norm on $x = y$

対角面 $x = y$ 上における熱流束密度 f のベクトルプロットです。鉄製の基盤底面は熱的に遮断されているため、底面からの熱の流出はありません。


(6) Contour(temp) on $y = 0$

平面 $y = 0$ 上での等温線図です。左右端は 300°K という境界条件を満たしています。




(7) Contour(temp) on $y = L/4$

円柱の中心点を通る平面 $y = L/4$ 上での等温線図です。


(8) Contour(temp) on $x = L/4$

円柱の中心点を通る平面 $x = L/4$ 上での等温線図です。鉄板部分における等温線図のパターンがプロット (7) と異なるのは $y = 0$ 面 (左端) が絶縁壁であるためです。




(9) Contour(temp) on $z = z_0$ on 'iron'

鉄製の基盤部上面における等温線図です。3つの側面は 300°K に保たれていますが、正面は絶縁壁であるため、等温線は直交する形になっています。


(10) Contour(temp) on $z = 0$

鉄製の基盤部底面における等温線図です。プロット (9) と同様の形状となっています。



2 Case2 - 時間変化

Case1 と基本的には同一の問題設定ですが、ここでは円柱頂部に定量の熱流を与え始めた時点からの時間変化を FlexPDE を使って考察してみます。対角面 $x = y$ 上における等温線図を 0.1 秒後から 1000 秒後までの範囲でプロットしてみます。十分な時間が経過した後の等温線図は平衡状態のそれに漸近して行くはずですが、時間変化の解析には数分の計算時間を要します。



2.1 Problem descriptor [3dfields2b.pde]

まずタイトルを設定します。

```
TITLE
  'Transient Heat Conduction'    { 3dfields2b.pde }
```

次に演算精度に関するセレクトをセットします。

```
SELECT
  Errlim = 1e-3
```

次に座標系が 3 次元直交座標系であることを明示します (デフォルトは Cartesian2)。

```
COORDINATES
  Cartesian3
```

従属変数を定義します。

```
VARIABLES
  temp
```

関連するパラメータや数式を定義します。

DEFINITIONS

```
L = 0.1  r0 = 0.01  z0 = 0.02
k  rcp  heat = 0          { Thermal parameters }
in_fluxd = 1e6           { Input heat flux density at top }
fluxd_x = -k*dx(temp)   fluxd_y = -k*dy(temp)   fluxd_z = -k*dz(temp)
fluxd = vector(fluxd_x, fluxd_y, fluxd_z)
fluxdm = magnitude(fluxd)
f_angle = sign(fluxd_y)*arccos(fluxd_x/fluxdm)/PI*180
```

初期状態を定義します。当初温度は一律 300°K の状態にあるとします。

INITIAL VALUES

```
temp = 300
```

時間依存型の熱伝導方程式は

$$\nabla \cdot \mathbf{f} + \rho c_p \frac{\partial T}{\partial t} = h$$

で与えられますが、背景情報については GB005 を参照ください。なお、 ρ (密度)、 c_p (比熱) の値は材質ごとに異なる点に注意する必要があります。

EQUATIONS

```
div(fluxd) + rcp*dt(temp) = heat
```

z 軸方向への extrusion を定義します。

```
EXTRUSION { Parallel surfaces }
Surface 'bottom' z = 0
Layer 'iron'
Surface 'middle' z = z0 { Interface }
Layer 'copper'
Surface 'top' z = L
```

BOUNDARIES セクションにおいて base plane 上でのリージョン構成と境界条件を定義します。

```
BOUNDARIES
Surface 'bottom' Natural(temp) = 0          { Insulated }
Surface 'top' Natural(temp) = -in.fluxd    { Input flux density }

Region 'domain'                             { Full solution domain }
  Layer 'iron' k = 82 rcp = 7.87e3*449     { Fe }
  Layer 'copper' void
  Start 'outer' (0, 0)
    Natural(temp) = 0 Line to (L, 0)
    Value(temp) = 300 Line to (L, L) to (0, L) to Close

Region 'cylinder'
  Layer 'iron' k = 82 rcp = 7.87e3*449     { Fe }
  Layer 'copper' k = 400 rcp = 8.96e3*385 { Redefine void to Cu }
  Start (L/4+r0, L/4) Arc(Center = L/4, L/4) Angle = 360 to Close
```

時間依存型問題に固有の指定項ですが、TIME セクションを用いて解析対象の時間帯を指定します。

```
TIME
0 to 1e3
```

最後に出力すべき情報を指定します。ここでは対角面 $x = y$ 上における等温線図を 0.1 秒後から 1000 秒後までの範囲でプロットします。

```
PLOTS
for t = 0.1, 5, 10, 15, 20, 25, 30, 35, 40, 45, 50, 55, 60, 100,
      200, 500, 1000
Contour(temp) painted on x = y
```

```
END
```


2.2 実行結果

上記スクリプトをそのまま FlexPDE に実行させるとすべてのプロットが一気に生成され、 $t = 1000$ に対応したプロットが表示された状態で処理が終了します。最初からすべてのプロットを確認するには「File」メニュー：「View File」と操作し、該当する .pg5 ファイルを選択してください。「View」メニュー：「Next」あるいは「Back」と操作することにより手動で前後に進めることができます。一方、「View」メニュー：「Movie」と操作すれば画面切替えは一定時間間隔で自動的に行われるようになります。

合計 17 枚のプロットが生成されますが、ここでは $t = 0.1, 60, 1000$ の 3 ケースについてのみ結果を示すことにします。


(1) Contour(temp) painted on $x = y$ at $t = 0.1$

対角面 $x = y$ 上での等温線図です。まだ時間が 0.1 秒しか経過していないため、円柱頂部を除けば温度は一律 300°K です。




(2) Contour(temp) painted on $x = y$ at $t = 60$

$t = 60$ における等温線図です。円柱頂部の温度はまだ 530°K であり、平衡状態には達していません。



(3) Contour(temp) painted on $x = y$ at $t = 1000$

$t = 1000$ における等温線図です。Case1 のプロット (3) と対比してみてください。



export2b: Cycle=124 Time=1000.0 dt=187.33 p2 Nodes=7989 Cells=4892 RMS En= 3.e-4
Integral= 1.701785

■